

Analyzing Restricted Mean Survival Time Using SAS/STAT®

Changbin Guo and Yu Liang, SAS Institute Inc., Cary, NC

ABSTRACT

Survival analysis handles time-to-event data. Classical methods, such as the log-rank test and the Cox proportional hazards model, focus on the hazard function and are most suitable when the proportional hazards assumption holds. When it does not hold, restricted mean survival time (RMST) methods often apply. The RMST is the expected survival time subject to a specific time horizon, and it is an alternative measure to summarize the survival profile. RMST-based inference has attracted attention from practitioners for its capability to handle nonproportionality. This paper introduces RMST methods in SAS/STAT® software: you can now use the RMSTREG procedure to fit linear and log-linear models, and you can use the RMST option in PROC LIFETEST to estimate the restricted mean survival time and make comparisons between groups. The paper discusses the rationale behind the RMST-based approach, outlines its recent development, and uses examples to illustrate real-world applications of the RMSTREG and LIFETEST procedures.

INTRODUCTION

The proportional hazards (PH) assumption plays an important role in survival data analysis. It is the basis of the popular Cox proportional hazards model. The widely applied log-rank test is equivalent to a score test of the PH model and achieves its highest power when the PH assumption is satisfied. However, practitioners have encountered various situations in which the PH assumption is questionable or simply untenable. Although it is possible to modify the classical methods to accommodate certain non-PH settings, solutions are rather limited, and often they are formulated at the expense of meaningful interpretations (Uno et al. 2014).

There are a handful of assumption-free measures of the survival outcome. For example, the mean and the median of the event time have been commonly used as summary statistics, and the differences or ratios of these quantities are often used to measure group effects. Censoring has posed special challenges that limit their usages. For example, if the last observation is censored, then you cannot reliably estimate the mean; and when not enough events occur, the median can be inestimable. The survival rate at a specified time is another commonly used summary statistic. Although it can be less problematic than the mean or median, it does not provide an overall summary of the time-to-event outcome.

The restricted mean survival time (RMST), sometimes called the restricted mean event time, is an alternative measure that is more often reliably estimable than the mean and median of the event time in certain situations. Also, it provides a summary of the whole survival curve up to a time horizon, in contrast to the survival rate at a specified time (Royston and Parmar 2013; Uno et al. 2014; Trinquart et al. 2016). The RMST has attracted practitioners for its straightforward interpretation and its capability to deal with nonproportional hazards. When two survival curves cross, the difference in the RMST between two groups still provides information about efficacy in a clinical trial, whereas the log-rank test fails to detect the significance and the hazard ratio becomes meaningless.

Starting in SAS/STAT 15.1, new, dedicated features are available for analyzing the RMST. You can use the RMST option in the LIFETEST procedure to perform nonparametric analysis with respect to the RMST. You can also use the new RMSTREG procedure to fit linear and log-linear models of the RMST.

This paper first presents the limitations of the classical hazard-based methods. It then introduces the RMST as an alternative measure and uses real-world data examples to illustrate how to perform nonparametric analysis with respect to the RMST. The next four sections focus on regression modeling and show how you can use pseudo-value regression and inverse probability censoring weighting techniques to fit the linear and log-linear models and make model-based inferences. Differences and connections with classical methods are also discussed.

LIMITATIONS OF HAZARD-BASED METHODS

In practical survival analysis, methods such as the log-rank test and Cox regression have become standard tools. Their use is warranted when the assumption of proportional hazards (PH) holds. In this case, the log-rank test is more powerful than other nonparametric tests, and the estimated effects in a Cox regression can be interpreted in terms of hazard ratios. However, the PH assumption is often not satisfied in real-world applications, and this can make the results unusable.

A typical violation of the PH assumption occurs when the two survival curves cross. This implies that the corresponding hazard functions would also cross. Because the log-rank statistic is essentially a weighted sum of the hazard function over time, the log-rank test loses much of its power to detect the true difference in survival. Results from the Cox PH regression have the same problem. As the true hazard ratio changes over time, the estimated hazard ratio from the fitted model ends up being a weighted average of the time-varying hazard ratios and can be interpreted as such. The problem is that the weights depend on the underlying survival and censoring distributions and therefore cannot be generalized straightforwardly.

RESTRICTED MEAN SURVIVAL TIME AS AN ALTERNATIVE MEASURE

Definition

Let T be a nonnegative random variable that represents the failure time of an individual from a homogeneous population. The survivor function (also known as the survival function) of T is defined as

$$S(t) = \Pr(T > t)$$

Assume that τ is a prespecified time point of interest. Let R be the minimum of T and τ :

$$R = T \wedge \tau = \min(T, \tau)$$

The restricted mean survival time is defined as the expected value of R :

$$\text{RMST}(\tau) = E(R) = E[\min(T, \tau)]$$

It can be evaluated by the area under the survivor function over $[0, \tau]$ as

$$\text{RMST}(\tau) = \int_0^{\tau} S(u) du$$

The restricted mean time lost (RMTL) is defined as the expected value of $\tau - R$:

$$\text{RMTL}(\tau) = E(\tau - R) = \tau - E[\min(T, \tau)] = \int_0^{\tau} [1 - S(u)] du$$

Statistical Analysis and Software

You can essentially make the same types of statistical inferences on the RMST as you can in the classical setting where the survival and hazard functions are the primary interests. Table 1 summarizes various inference objectives in the classical setting and the RMST setting.

Table 1 Inference Objectives in Classical and RMST Settings

Task	Classical Setting	RMST Setting
Estimation	$S(t)$	$\text{RMST}(\tau)$
Testing	$H_0: S_1(t) = S_2(t)$ $H_1: S_1(t) \neq S_2(t)$	$H_0: \text{RMST}_1(\tau) = \text{RMST}_2(\tau)$ $H_1: \text{RMST}_1(\tau) \neq \text{RMST}_2(\tau)$
Regression	PH: $h(t) = h_0(t) \exp(\mathbf{x}'\boldsymbol{\beta})$ AFT: $\log(t) = \beta_0 + \mathbf{x}'\boldsymbol{\beta} + \epsilon$	Linear: $\text{RMST}(\tau) = \beta_0 + \mathbf{x}'\boldsymbol{\beta}$ Log-linear: $\log(\text{RMST}(\tau)) = \beta_0 + \mathbf{x}'\boldsymbol{\beta}$

SAS/STAT software is well developed in the classical setting. PROC LIFETEST traditionally focuses on the estimating and testing tasks for the survival functions and supports nonparametric methods such as the Kaplan-Meier estimator

and the log-rank test. The assumption-free nonparametric methods for the RMST extend these classical methods. In SAS/STAT 15.1, you can use the new RMST option in the LIFETEST procedure to estimate and compare the RMST.

The proportional hazards (PH) model and the accelerated failure time (AFT) model are popular choices for analyzing time-to-event data. In SAS/STAT, the PHREG procedure fits primarily the Cox PH model to right-censored data but also fits other types of PH models. The LIFEREG procedure can fit parametric AFT models to arbitrarily censored data. A key difference between the two procedures is that the Cox PH model does not assume a particular form on the baseline hazard function $h_0(t)$, whereas the AFT model assumes that the error term ϵ follows certain parametric distributions, such as Weibull or exponential. The new RMSTREG procedure in SAS/STAT 15.1 provides regression modeling capabilities for the RMST setting and fits generalized linear models such as linear and log-linear models to right-censored data. Table 2 summarizes the key features of these procedures.

Table 2 Survival Modeling Procedures

Procedure	Focus	Model Type	Estimation Method
PROC LIFEREG	Time to event	Accelerated failure time models	Likelihood
PROC PHREG	Hazard function	Proportional hazards models	Partial likelihood
PROC RMSTREG	Restricted mean survival time	Generalized linear models	Estimating equations

It is possible to estimate RMST from a classical survival model, such as the Cox proportional hazards model (Zucker 1998), but the process is complex and difficult to extend. PROC RMSTREG avoids this difficulty by using generalized linear modeling techniques to directly model the RMST. This approach has the double advantage of making inferences on the results straightforward and providing all the machinery of generalized linear model comparisons for studying RMST effects.

NONPARAMETRIC ANALYSIS USING PROC LIFETEST

The new methods of analyzing the RMST in the LIFETEST procedure focus on nonparametric estimation and group comparisons of the RMST. This section briefly presents these methods.

Estimating the RMST

Let $t_1 < t_2 < \dots < t_D$ represent distinct event times. For each $i = 1, \dots, D$, let Y_i be the number of surviving units (the size of the risk set) just prior to t_i , and let d_i be the number of units that fail at t_i .

The Kaplan-Meier (product-limit) estimate of the survivor function at t_i is the cumulative product

$$\hat{S}(t_i) = \prod_{j=1}^i \left(1 - \frac{d_j}{Y_j}\right)$$

If the largest observed time is uncensored, the estimated mean survival time is

$$\hat{\mu} = \sum_{i=1}^D \hat{S}(t_{i-1})(t_i - t_{i-1})$$

where t_0 is defined to be 0.

RMST(τ) is estimated by

$$\widehat{\text{RMST}}(\tau) = \int_0^{\tau} \hat{S}(t) dt = \sum_{i=1}^{N^*} \hat{S}(t_{i-1})(t_i - t_{i-1}) + \hat{S}(t_{N^*})(\tau - t_{N^*})$$

where N^* is the number of t_i values that are less than τ .

The restricted mean time lost (RMTL) is estimated by $\widehat{\text{RMTL}}(\tau) = \tau - \widehat{\text{RMST}}(\tau)$.

The standard error of $\widehat{\text{RMST}}(\tau)$ or $\widehat{\text{RMTL}}(\tau)$ is estimated as

$$\hat{\sigma} = \sqrt{\frac{m}{m-1} \sum_{i=1}^{N^*} \frac{d_i A_i^2}{Y_i(Y_i - d_i)}}$$

where

$$A_i = \int_{t_i}^{\tau} \hat{S}(t) dt = \sum_{j=i}^{N^*} \hat{S}(t_j)(t_{j+1} - t_j) + \hat{S}(t_{N^*})(\tau - t_{N^*})$$

$$m = \sum_{j=1}^{N^*} d_j$$

Comparing the RMST between Groups

Let K be the number of groups. Let $S_k(t)$ be the underlying survivor function of the k th group, $k = 1, \dots, K$.

Assume that τ is a prespecified time point of interest and $S_k(\tau) > 0$. The following methods are presented in terms of $\text{RMST}(\tau)$, but they also extend to the analysis of $\text{RMSTL}(\tau)$.

The null and alternative hypotheses to be tested are

$$H_0 : \text{RMST}_1(\tau) = \text{RMST}_2(\tau) = \dots = \text{RMST}_K(\tau)$$

versus

$$H_1 : \text{at least one of the } \text{RMST}_k(\tau)\text{'s is different}$$

Let $\widehat{\text{RMST}}(\tau) = [\widehat{\text{RMST}}_1(\tau), \widehat{\text{RMST}}_2(\tau), \dots, \widehat{\text{RMST}}_K(\tau)]^T$ be the vector of estimated RMSTs for the K groups.

Let $\hat{\Sigma}$ be the estimated covariance matrix for $\widehat{\text{RMST}}(\tau)$. It is a diagonal matrix, and the j th diagonal element is $\hat{\sigma}_j^2$, which is the estimated variance of $\widehat{\text{RMST}}_j(\tau)$.

Let D be a $(K - 1) \times K$ matrix whose j th row is $e_j - e_{j+1}$, where e_j is a K -dimensional vector whose j th element is 1 and whose other elements are 0. The test statistic is computed as

$$(\widehat{\text{RMST}}(\tau))^T D^T (D \hat{\Sigma} D^T)^{-1} D \widehat{\text{RMST}}(\tau)$$

Under the null hypothesis H_0 , this K -sample test statistic has approximately a chi-square distribution with degrees of freedom equal to the rank of $D \hat{\Sigma} D^T$.

EXAMPLE OF NONPARAMETRIC ANALYSES

This example uses real-world data to demonstrate how you can perform nonparametric analyses of the RMST. The data, which are presented in Appendix I of Kalbfleisch and Prentice (1980), are coded in the following DATA step. The response variable, **SurvTime**, is the survival time in days of a lung cancer patient. The covariates are **Cell** (type of cancer cell), **Therapy** (type of therapy: standard or test), **Prior** (prior therapy: 0=no, 1=yes), **Age** (age in years), **DiagTime** (time in months from diagnosis to entry into the trial), and **Kps** (performance status). The censoring indicator variable **Censor** is created from the data; the value 1 indicates a censored time, and the value 0 indicates an event time. Because there are only two types of therapy, an indicator variable, **Treatment**, is created for therapy type; the value 0 indicates standard therapy, and the value 1 indicates test therapy.

```
data VALung;
  drop check m;
  retain Therapy Cell;
  infile cards column=column;
  length Check $ 1;
  label SurvTime='Failure or Censoring Time'
        Kps='Karnofsky Index'
        DiagTime='Months till Randomization'
        Age='Age in Years'
        Prior='Prior Treatment?'
        Cell='Cell Type'
        Therapy='Type of Treatment'
        Treatment='Treatment Indicator';
```

```

... more lines ...

52 60 4 45 0 164 70 15 68 10 19 30 4 39 10 53 60 12 66 0
15 30 5 63 0 43 60 11 49 10 340 80 10 64 10 133 75 1 65 0
111 60 5 64 0 231 70 18 67 10 378 80 4 65 0 49 30 3 37 0
;

```

The following statements use PROC LIFETEST to perform analyses of the restricted mean survival time (RMST) and restricted mean time lost (RMTL) in addition to the standard analyses:

```

ods graphics on;
proc lifetest data=VALung plots=(rmst rmtl s) rmst rmtl(tau=90) maxtime=600;
time SurvTime*Censor(1);
strata Cell;
run;
ods graphics off;

```

The RMST and RMTL options estimate the restricted mean survival time and the restricted mean time lost, respectively. The variable **Cell** is specified in the STRATA statement to compute the RMST for each type of cancer cell. ODS Graphics must be enabled for graphs to be produced. Graphical displays of the RMST and RMTL curves are requested through the PLOTS= option in the PROC LIFETEST statement. Because of a few large survival times, a MAXTIME= option value of 600 is used to set the upper limit of the time axis; that is, the time horizon in the plots extends from 0 to a maximum of 600 days.

The graph of the Kaplan-Meier curves is shown in [Figure 1](#).

Figure 1 Survival Curves




Figure 2 displays the τ value that the RMST analysis uses. If you omit the TAU= option, PROC LIFETEST uses the smallest value among the largest observed times across the strata as the τ value.

Figure 2 RMST Analysis Information

RMST Analysis Information	
Tau	186

Figure 3 displays the RMST estimates for the four cell types.

Figure 3 RMST Estimates

RMST Estimates			
Stratum	Cell Type	Estimate	Standard Error
1	adeno	65.55556	9.9303
2	large	128.0370	11.9858
3	small	64.20647	8.2859
4	squamous	113.8040	12.3703

Figure 4 displays information about the RMTL analysis. A τ value of 90 is shown; this is the value that is specified in the TAU= option.

Figure 4 RMTL Analysis Information

RMTL Analysis Information	
Tau	90


Figure 5 displays the RMTL estimates for the four cell types at $\tau = 90$.

Figure 5 RMTL Estimates at $\tau = 90$

RMTL Estimates			
Stratum	Cell Type	Estimate	Standard Error
1	adeno	36.54321	6.3206
2	large	14.33333	4.9560
3	small	41.27083	4.6763
4	squamous	22.99365	5.6430


The graph of the estimated RMST curves is shown in Figure 6. These curves exhibit a behavior similar to that of the survival curves in Figure 1: the adeno cell curve and the small cell curve are much closer to each other than they are to the large cell curve or the squamous cell curve. The shapes of the large cell curve and the squamous cell curve are quite different, although the RMST in both curves increases more rapidly than the RMST in the adeno and small cell curves. The RMST in the squamous cell curve initially increases less rapidly than the RMST in the large cell curve, but the roles are reversed in the later period.

Figure 6 RMST Curves



The graph of the estimated RMTL curves is displayed in Figure 7. Again, the adeno cell curve and the small cell curve are much closer to each other and farther away from the large cell and squamous cell curves.

Figure 7 RMTL Curves



Results of the homogeneity test for the RMST across cell types are given in Figure 8. The table displays the approximate chi-square statistic, degrees of freedom, and p -value. The test results provide strong evidence that the RMSTs for the four types of cancer cells are not the same ($p < 0.0001$).

Figure 8 Homogeneity Tests across Cell Types

RMST Test of Equality			
Source	Chi-Square	DF	Pr > ChiSq
Strata	28.4427	3	<.0001

The homogeneity test does not identify which pairs of the RMSTs are different. In the following statements, you use the DIFF= option to compute the paired differences of the RMST among the groups. To protect yourself from falsely significant results, you use the ADJ= option to make multiple-comparison adjustments to the resulting p -values.

```
proc lifetest data=VALung rmst;
  time SurvTime*Censor(1);
  strata Cell / diff=all adj=sidak;
run;
```

The Šidák multiple-comparison results are shown in Figure 9.

Figure 9 All Paired Comparisons
The LIFETEST Procedure

Restricted Mean Survival Time Comparisons					
Adjustment for Multiple Comparisons: Sidak					
Pr > ChiSq					
Stratum Comparison	Difference	Standard Error	Chi-Square	Unadjusted	Adjusted
adeno large	-62.4815	15.5650	16.1141	<.0001	0.0004
adeno small	1.349087	12.9332	0.0109	0.9169	1.0000
adeno squamous	-48.2485	15.8630	9.2511	0.0024	0.0140
large small	63.83057	14.5710	19.1901	<.0001	<.0001
large squamous	14.23303	17.2245	0.6828	0.4086	0.9572
small squamous	-49.5975	14.8889	11.0967	0.0009	0.0052

The results suggest that you can divide the four risk groups into two classes. The first class consists of the small and adeno cell types, and there is no significant difference in the RMST between them ($p = 1.0000$). The second class consists of the large and squamous cell types, and the paired comparison is not significant ($p = 0.9572$). However, there is a significant difference in any paired comparison between the two classes.

Suppose you consider the small cell type to be the reference group. You can use the DIFF= option in the STRATA statement to designate this risk group as the control and apply a multiple-comparison adjustment to the p -values for the paired comparison between the small cell type and the other cell types. Consider the Šidák correction again. You specify the ADJ= and DIFF= options as in the following statements:

```
proc lifetest data=VALung rmst;
  time SurvTime*Censor(1);
  strata Cell / adj=sidak diff=control('small');
run;
```

Figure 10 shows that both the large and squamous cell types differ from the small cell type at the 0.05 significance level, whereas the difference between the adeno and small cell types is not significant ($p = 0.9994$). As you can see, in a period of 186 days, patients with the large cell type tend to live 63.8 days longer than those with the small cell type.

Figure 10 Comparisons to the Reference Group
The LIFETEST Procedure

Restricted Mean Survival Time Comparisons					
Adjustment for Multiple Comparisons: Sidak					
Pr > ChiSq					
Stratum Comparison	Difference	Standard Error	Chi-Square	Unadjusted	Adjusted
adeno small	1.349087	12.9332	0.0109	0.9169	0.9994
large small	63.83057	14.5710	19.1901	<.0001	<.0001
squamous small	49.59754	14.8889	11.0967	0.0009	0.0026

REGRESSION ANALYSIS USING PROC RMSTREG

Regression analyses aim to study the relationship between the dependent variable and the independent variables. Although nonparametric methods are usually your first resort in analyzing the RMST, regression modeling can be more effective in describing the underlying relationship between the RMST and independent variables. The linear model of the RMST is the simplest formulation that you can consider as a generalization of the simple linear regression to the RMST. Because the RMST is nonnegative, the linear model might go out of bounds, and this can compromise its interpretation. A natural alternative to the linear model is the log-linear model, which instead models the natural log of the RMST. For the linear model, you can interpret the effects as differences in the RMST. For the log-linear model, you

can interpret the effects in terms of ratios in the RMST. Both linear and log-linear models are special cases of the generalized linear models (Nelder and Wedderburn 1972).

Let D_i be the response variable for the i th observation. The quantity \mathbf{x}_i is a column vector of covariates, or explanatory variables, for observation i that is known from the experimental setting and is considered to be fixed, or nonrandom.

The expected value of D_i , denoted by μ_i , is

$$\mu_i = E[D_i | \mathbf{x}_i]$$

Under the specification of generalized linear models (Nelder and Wedderburn 1972), μ_i is related to a linear predictor through a monotone and differentiable link function g ,

$$g(\mu_i) = \mathbf{x}_i' \boldsymbol{\beta}$$

where $\boldsymbol{\beta}$ is an unknown parameter vector and another column can be added to \mathbf{x}_i for an intercept effect.

Assume that τ is a prespecified time point of interest. Let T_i be the time-to-event variable for the i th subject. The subject-specific RMST at τ is defined by $\text{RMST}_i(\tau) = E[\min(T_i, \tau)]$ and can be conveniently modeled via a generalized linear model as

$$g[\text{RMST}_i(\tau)] = \mathbf{x}_i' \boldsymbol{\beta}$$

Under the natural logarithm link $g(\cdot) = \log(\cdot)$, the model is

$$\log[\text{RMST}_i(\tau)] = \mathbf{x}_i' \boldsymbol{\beta}$$

Under the identity or linear link, the model is

$$\text{RMST}_i(\tau) = \mathbf{x}_i' \boldsymbol{\beta}$$

The RMSTREG procedure analyzes time-to-event data by using regression with respect to the restricted mean survival time. It provides you with two methods to fit these models: the pseudo-value regression method (Andersen, Hansen, and Klein 2004) and the inverse probability censoring weighting (IPCW) method (Tian, Zhao, and Wei 2014). Both methods are generic approaches for fitting regression models to censored data, and you can apply them to the RMST setting. But the methods have their differences. The pseudo-value regression method assumes that only the censoring is noninformative to the survival outcomes and treats the censoring distribution as a nuisance in the estimation. On the other hand, the IPCW technique further assumes that the censoring distribution can be properly estimated and uses the estimates as weights explicitly in the estimation process. This feature allows the IPCW method to deal with nonhomogeneous censoring, so it can be more efficient in those settings. However, the downside is that the IPCW method requires the censoring distribution to be correctly estimated. The following two sections discuss these two estimation methods in detail and present examples to illustrate how you can use them in PROC RMSTREG.

PSEUDOVALUE REGRESSION

Pseudo-value regression is a generic method of fitting generalized linear models to time-to-event data (Andersen, Klein, and Rosthøj 2003). This section describes how the method works and how you can apply it to analyze models of the RMST.

Let $\mathbf{D}_1, \dots, \mathbf{D}_n$ be independent and identically distributed quantities that might be random variables or vectors of variables. Let $\boldsymbol{\theta} = E[f(\mathbf{D}_i)]$ for some function $f(\cdot)$. Suppose that $\hat{\boldsymbol{\theta}}$ is an unbiased estimator of $\boldsymbol{\theta}$.

Let $\mathbf{x}_1, \dots, \mathbf{x}_n$ be independent and identically distributed samples of covariates, and define the conditional expectation of $f(\mathbf{D}_i)$ given by \mathbf{x}_i as

$$\boldsymbol{\theta}_i = E[f(\mathbf{D}_i) | \mathbf{x}_i]$$

The i th pseudo-observation of $\boldsymbol{\theta}$ is computed as

$$\hat{\boldsymbol{\theta}}_i = n\hat{\boldsymbol{\theta}} - (n-1)\hat{\boldsymbol{\theta}}^{-i}$$

where $\hat{\boldsymbol{\theta}}^{-i}$ is the jackknife leave-one-out estimator for $\boldsymbol{\theta}$ based on $\{\mathbf{D}_j : j \neq i\}$.

The generalized linear model (Nelder and Wedderburn 1972) for $\boldsymbol{\theta}$ assumes

$$g(\boldsymbol{\theta}_i) = \mathbf{x}_i' \boldsymbol{\beta}$$

where $g(\cdot)$ is a suitable link function. Note that another column can be added to X_i for an intercept effect.

Using pseudo-observations, you can estimate the regression parameters β by solving the following estimating equations,

$$\mathbf{u}(\beta) = \sum_{i=1}^n \mathbf{u}_i(\beta) = \sum_{i=1}^n \left(\frac{\partial \theta_i}{\partial \beta} \right)' \mathbf{V}_i^{-1} (\hat{\theta}_i - \theta_i) = \mathbf{0}$$

where \mathbf{V}_i is a working covariance matrix.

Let $\hat{\beta}$ be the solution of the estimating equations. You can use a sandwich estimator to estimate the variance-covariance matrix of $\hat{\beta}$. It takes the form

$$\Sigma_e = \mathbf{I}_0^{-1} \mathbf{I}_1 \mathbf{I}_0^{-1}$$

\mathbf{I}_0^{-1} is the model-based estimator of $\text{Cov}(\hat{\beta})$ and is given by

$$\mathbf{I}_0 = \sum_{i=1}^n \frac{\partial \theta_i}{\partial \beta} \left(\frac{\partial \theta_i}{\partial \beta} \right)' \mathbf{V}_i^{-1} \frac{\partial \theta_i}{\partial \beta}$$

\mathbf{I}_1^{-1} is the empirical estimator of $\text{Cov}(\hat{\beta})$ and is computed as

$$\mathbf{I}_1 = \sum_{i=1}^n \mathbf{u}_i(\hat{\beta}) \mathbf{u}_i(\hat{\beta})'$$

Andersen, Hansen, and Klein (2004) proposed using pseudo-value regression to analyze RMST models. Assume that τ is a prespecified time point of interest. Let T_i be the time-to-event variable for the i th subject. You can use pseudo-value regression to fit the RMST models by letting

$$\theta_i = \text{RMST}_i(\tau) = E(T_i \wedge \tau | \mathbf{x}_i)$$

$$\mathbf{V}_i = \begin{cases} \theta_i = \text{RMST}_i(\tau) & g(u) = \log(u) \\ 1 & g(u) = u \end{cases}$$

Because the nonparametric estimator $\widehat{\text{RMST}}(\tau)$ based on the Kaplan-Meier estimator is unbiased, you can use it in place of $\hat{\theta}$ in the estimation process.

Example of Pseudo-value Regression

The data in this example represent 418 patients who have primary biliary cirrhosis (PBC), among whom 161 had died as of the date of the data listing. A subset of the variables is saved in the SAS data set **Liver**. The data set contains the following variables:

- **Time**, follow-up time, in years
- **Status**, event indicator, with the value 1 for death time and 0 for censored time
- **Age**, age in years, from birth to study registration
- **Albumin**, serum albumin level, in g/dl
- **Bilirubin**, serum bilirubin level, in mg/dl
- **Edema**, edema presence
- **Prottime**, prothrombin time, in seconds

The following statements create the data set **Liver**:

```
data Liver;
  input Time Status Age Albumin Bilirubin Edema Prottime @@;
  label Time="Follow-Up Time in Years";
  Time= Time / 365.25;
  datalines;
  400 1 58.7652 2.60 14.5 1.0 12.2 4500 0 56.4463 4.14 1.1 0.0 10.6
  1012 1 70.0726 3.48 1.4 0.5 12.0 1925 1 54.7406 2.54 1.8 0.5 10.3
  1504 0 38.1054 3.53 3.4 0.0 10.9 2503 1 66.2587 3.98 0.8 0.0 11.0
  1832 0 55.5346 4.09 1.0 0.0 9.7 2466 1 53.0568 4.00 0.3 0.0 11.0
  2400 1 42.5079 3.08 3.2 0.0 11.0 51 1 70.5599 2.74 12.6 1.0 11.5
  3762 1 53.7139 4.16 1.4 0.0 12.0 304 1 59.1376 3.52 3.6 0.0 13.6
  3577 0 45.6893 3.85 0.7 0.0 10.6 1217 1 56.2218 2.27 0.8 1.0 11.0
  3584 1 64.6461 3.87 0.8 0.0 11.0 3672 0 40.4435 3.66 0.7 0.0 10.8

  ... more lines ...

  989 0 35.0000 3.23 0.7 0.0 10.8 681 1 67.0000 2.96 1.2 0.0 10.9
  1103 0 39.0000 3.83 0.9 0.0 11.2 1055 0 57.0000 3.42 1.6 0.0 9.9
  691 0 58.0000 3.75 0.8 0.0 10.4 976 0 53.0000 3.29 0.7 0.0 10.6
;
```

The following statements fit a linear model of the RMST with the covariates **Bilirubin**, **Age**, and **Edema**:

```
proc rmstreg data=liver tau=10;
  class Edema;
  model Time*Status(0) = Age Bilirubin Edema / link=linear method=pv;
run;
```

The **TAU=** option in the PROC RMSTREG statement specifies the time limit that defines the RMST for this analysis. If you omit this option, the largest observed time from the input data is used. Because the variable **Edema** is specified as a CLASS variable, it contributes one dummy variable to the regression model for each of its values.

In the MODEL statement, the response consists of the observed variable **Time** and an indicator variable **Status**, which specifies whether or not the **Time** value is censored. The values of **Time** are considered to be censored if the value of **Status** is 0; otherwise, they are considered to be event times.

An intercept term is included by default. Thus, the model matrix **X** consists of a column of 1s that represent the intercept term, three columns of 0s and 1s that correspond to the levels of the **Edema** variable, and two additional columns for the values of the variables **Age** and **Bilirubin**.

That is, the model matrix is

$$\mathbf{X} = \left[\begin{array}{c|ccc|cc} 1 & 1 & 0 & 0 & \text{Age} & \text{Bilirubin} \\ 1 & 0 & 1 & 0 & \text{Age} & \text{Bilirubin} \\ 1 & 0 & 0 & 1 & \text{Age} & \text{Bilirubin} \end{array} \right]$$

The **LINK=LINEAR** option fits a linear model. That is, the RMST at $\tau = 10$ for a specific subject (denoted by μ_i) is related to the linear predictor by

$$\mu_i = \mathbf{x}'_i \boldsymbol{\beta}$$

The **METHOD=PV** option specifies pseudo-value regression to fit the model.

Figure 11 shows the “Model Information” table, which provides information about the specified linear model of the RMST and the input data set.

Figure 11 Model Information
The RMSTREG Procedure

Model Information	
Data Set	WORK.LIVER
Time Variable	Time
Censoring Variable	Status
Censoring Value(s)	0
Link Function	Linear
Estimation Method	Pseudo Value
Tau Value	10

Figure 12 displays a summary of the number of event and censored observations in the data set.

Figure 12 Event and Censoring Summary

Summary of the Number of Event and Censored Values		
Total	Event	Censored
418	161	257

Figure 13 shows how the **Edema** variable is coded in the model matrix.

Figure 13 CLASS Variable Level Information

Class Level Information				
Class	Value	Design Variables		
		Edema 0	1	0
0.5	0	1	0	
1	0	0	1	

For each parameter in the model, PROC RMSTREG displays a table (Figure 14) that contains columns of the parameter name, the degrees of freedom associated with the parameter, the estimated parameter value, the standard error of the parameter estimate, the confidence intervals, and the Wald chi-square statistic and associated p -value for testing the significance of the parameter to the model. If a column of the model matrix that corresponds to a parameter is found to be linearly dependent, or *aliased*, with columns that correspond to parameters preceding it in the model, PROC RMSTREG assigns it zero degrees of freedom and displays a value of 0 for the parameter estimate.

Figure 14 Analysis of Parameter Estimates

Analysis of Parameter Estimates								
Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq	
				Intercept	1			9.0588
Age	1	-0.0686	0.0142	-0.0964	-0.0408	23.41	<.0001	
Bilirubin	1	-0.3614	0.0383	-0.4366	-0.2862	88.82	<.0001	
Edema	0	1	3.0259	0.7353	1.5849	4.4670	16.94	<.0001
Edema	0.5	1	1.7778	0.8658	0.0807	3.4748	4.22	0.0401
Edema	1	0	0.0000					

By default, PROC RMSTREG performs a Type 3 analysis for assessing the significance of each model effect. The results are shown in Figure 15.

Figure 15 Type 3 Analysis of Effects

Type 3 Analysis of Effects			
Effect	DF	Chi-Square	Pr > ChiSq
Age	1	23.4052	<.0001
Bilirubin	1	88.8223	<.0001
Edema	2	19.4651	<.0001

The results indicate that all three variables are strong predictors of the RMST at $\tau = 10$. The Type 3 chi-square value for the **Age** variable, for example, is computed by comparing the model with **Intercept**, **Age**, **Bilirubin**, and **Edema** included to the model with **Age** excluded. The hypothesis that is tested in this case is that **Age** adds no predictability to the model, over and above **Edema** and **Bilirubin**. This hypothesis is strongly rejected.

If the RMST at $\tau = 5$ is of interest, then the following statements fit this model by using pseudo-value regression:

```
proc rmstreg data=liver tau=5;
  class Edema;
  model Time*Status(0) = Age Bilirubin Edema / link=linear method=pv;
run;
```

Figure 16 displays the parameter estimates and the Type 3 tests for the fitted model. The new results appear to be similar to the previous ones. Using values 5 and 10 for τ , respectively, makes little difference.

Figure 16 Parameter Estimates and Type 3 Tests for RMST at $\tau = 5$

The RMSTREG Procedure

Type 3 Analysis of Effects			
Effect	DF	Chi-Square	Pr > ChiSq
Age	1	18.9013	<.0001
Bilirubin	1	62.3617	<.0001
Edema	2	30.8671	<.0001

Analysis of Parameter Estimates

Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq	
Intercept	1	4.0394	0.4848	3.0892	4.9896	69.42	<.0001	
Age	1	-0.0220	0.0051	-0.0319	-0.0121	18.90	<.0001	
Bilirubin	1	-0.1378	0.0174	-0.1720	-0.1036	62.36	<.0001	
Edema	0	1	1.8901	0.3621	1.1805	2.5997	27.25	<.0001
Edema	0.5	1	1.2573	0.4205	0.4331	2.0815	8.94	0.0028
Edema	1	0	0.0000					

IPCW REGRESSION

IPCW regression offers an alternative to pseudo-value regression for fitting RMST models. The main difference is that IPCW regression assumes that the censoring distribution can be correctly estimated and pseudo-value regression does not. The standard application of IPCW regression assumes that censoring is homogeneous among subjects, so that you can use the Kaplan-Meier method to estimate the censoring distribution. When grouping is present, a straightforward extension is to allow the censoring to be homogeneous within groups. This section discusses how IPCW regression works and presents examples to illustrate how you can use it in PROC RMSTREG.

Denote T as the event time, C as the censoring time, and x as a p -dimensional vector of covariates. For the i th subject, $i = 1, \dots, n$, let $U_i = T_i \wedge C_i$, $\Delta_i = I(T_i \leq C_i)$, and x_i be the observed time, event indicator, and covariate vector, respectively.

Assume that τ is a prespecified time point of interest and $P(T > \tau) > 0$. Let

$$\begin{aligned} R_i &= T_i \wedge \tau \\ \text{RMST}_i(\tau) &= E(R_i | \mathbf{x}_i) \\ \tilde{\Delta}_i &= I(R_i \leq C_i) \\ w_i &= \frac{\tilde{\Delta}_i}{\hat{G}(R_i)} \end{aligned}$$

where $\hat{G}(t)$ is the Kaplan-Meier estimate (alternatively, the Breslow estimate) of the survival function of the censoring variable, which is calculated using $\{(U_i, 1 - \Delta_i) : i = 1, 2, \dots, n\}$.

Suppose that the following relationship holds for the RMST,

$$g[\text{RMST}_i(\tau)] = \mathbf{x}'_i \boldsymbol{\beta}$$

where $g(\cdot)$ is a smooth and strictly increasing function. Note that another column can be added to X_i for an intercept effect.

Under suitable regularity conditions, you estimate the regression coefficients $\boldsymbol{\beta}$ by solving the following score function (Tian, Zhao, and Wei 2014):

$$\mathbf{U}(\boldsymbol{\beta}) = \sum_{i=1}^n w_i \left(R_i - g^{-1}(\mathbf{x}'_i \boldsymbol{\beta}) \right) \mathbf{x}_i = \mathbf{0}$$

Let

$$\hat{\boldsymbol{\Omega}} = \sum_{i=1}^n \mathbf{x}_i \otimes^2 \left(g^{-1}(\mathbf{x}'_i \hat{\boldsymbol{\beta}}) \right)$$

The sandwich variance estimate of $\hat{\boldsymbol{\beta}}$ is

$$\widehat{\text{Var}}(\hat{\boldsymbol{\beta}}) = \hat{\boldsymbol{\Omega}}^{-1} \hat{\boldsymbol{\Sigma}} \hat{\boldsymbol{\Omega}}^{-1}$$

See the PROC RMSTREG documentation in the *SAS/STAT 15.1 User's Guide* for a complete presentation of the details.

The assumption of homogeneous censoring can be relaxed as follows. Assuming that you have K strata, within each stratum the censoring distribution is homogeneous. For the i th subject, let $B_i \in (1, \dots, K)$ be the stratum indicator. It is more appropriate to use stratum-specific weights in the estimation. For the k th stratum, you compute the Kaplan-Meier estimate $\hat{G}_k(t)$ for the censoring variable by using $\{(U_i, 1 - \Delta_i) : B_i = k, i = 1, 2, \dots, n\}$.

For the i th subject, the weight is computed as

$$w_i = \frac{\tilde{\Delta}_i}{\hat{G}_{k=B_i}(R_i)}$$

Example of IPCW Regression

In a study of the human immunodeficiency virus (HIV), patients were followed after a confirmed HIV-positive diagnosis (Hosmer and Lemeshow 1999). The primary goal is to evaluate the effect of two different covariates on mortality: the patient's age and the patient's history of intravenous drug use.

The following DATA step creates the data set **HIV**, which contains the variables **Time** (the follow-up time in days), **Status** (with a value of 0 if **Time** was censored and 1 otherwise), **Drug** (with a value of 1 for prior intravenous drug use and 0 otherwise), and **Age** (the patient's age in years at the beginning of the follow-up):

```
data HIV;
  input Time Age Drug Status;
  datalines;
  5      46      0      1
  6      35      1      0
  8      30      1      1
```

```

3      30      1      1
22     36      0      1
1      32      1      0
... more lines ...
1      34      1      1
;

```

The following statements fit the linear model for the RMST at $\tau = 48$ by using the method of inverse probability censoring weighting (IPCW):

```

proc rmstreg data=hiv tau=48;
  class Drug;
  model Time*Status(0) = Drug Age / link=linear method=ipcw;
run;

```

Figure 17 shows the “Model Information” table, which provides information about the specified linear model of the RMST and the input data set.

Figure 17 Model Information
The RMSTREG Procedure

Model Information	
Data Set	WORK.HIV
Time Variable	Time
Censoring Variable	Status
Censoring Value(s)	0
Link Function	Linear
Estimation Method	IPCW
Tau Value	48

Figure 18 displays the parameter estimates for the fitted linear model.

Figure 18 Parameter Estimates

Analysis of Parameter Estimates							
Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
				Intercept	1		
Drug	0	12.5483	2.6029	7.4468	17.6499	23.24	<.0001
Drug	1	0	0.0000				
Age	1	-1.0748	0.2289	-1.5234	-0.6262	22.05	<.0001

The results show that both age and prior drug use are strongly associated with mortality.

The standard IPCW method assumes homogeneity of the censoring mechanism. Such an assumption can be evaluated statistically. The following statements use PROC LIFETEST to estimate the survival functions of censoring for the two levels of **Drug** and use the log-rank test to test whether censoring is homogeneous between the two **Drug** groups. Note that the censoring value in the TIME statement uses 1 instead of 0.


```

ods graphics on;
proc lifetest data=hiv plot=s(test);
  strata Drug;
  time Time*Status(1);
run;

```

Figure 19 displays the Kaplan-Meier curves for the two **Drug** groups.

Figure 19 Plot of Estimated Survival Functions



The curves show substantial separation over time. The small p -value from the log-rank test also suggests that the censoring patterns of the two groups are quite different.

By default, the IPCW method uses the Kaplan-Meier technique to obtain the weights, and this approach implicitly assumes that the censoring mechanism is homogeneous among all subjects. Sometimes this is not a reasonable assumption—for example, when there are distinct groups, such as treatment arms in randomized clinical trials. Under such circumstances, it is more appropriate to use group-specific weights by applying the Kaplan-Meier technique to different groups.

The following statements fit the linear model for the RMST at $\tau = 48$ by using the IPCW method, with weights estimated separately for the two **Drug** groups:

```
proc rmstreg data=hiv tau=48;
  class Drug;
  model Time*Status(0) = Drug Age / link=linear method=ipcw(strata=Drug);
run;
```

Figure 20 displays the parameter estimates for the fitted linear model of the RMST. As you can see, the parameter estimates are similar to those displayed in Figure 18.

Figure 20 Parameter Estimates
The RMSTREG Procedure

Analysis of Parameter Estimates							
Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept	1	47.5159	10.5939	26.7522	68.2795	20.12	<.0001
Drug	0 1	11.4049	2.5667	6.3743	16.4354	19.74	<.0001
Drug	1 0	0.0000					
Age	1	-1.1380	0.2654	-1.6582	-0.6179	18.39	<.0001

PROC RMSTREG VERSUS OTHER PROCEDURES

When it comes to modeling time-to-event data, proportional hazards (PH) models and accelerated failure time (AFT) models are popular tools. You can use the PHREG and LIFEREG procedures, respectively, to fit them. Generalized linear models have become standard tools for analyzing categorical data and are well known for their flexibility. The RMSTREG procedure fits generalized linear models of the RMST by using specialized methods and provides useful alternatives to the PH and AFT models. The example that follows compares PROC RMSTREG with PROC PHREG and PROC LIFEREG by analyzing real-world data.

Krall, Uthoff, and Harley (1975) analyzed data from a study on multiple myeloma in which researchers treated 65 patients by using alkylating agents. Of those patients, 48 died during the study and 17 survived. The following DATA step creates the data set **Myeloma**:

```
data Myeloma;
  input Time VStatus LogBUN HGB Platelet Age LogWBC Frac
        LogPBM Protein SCalc;
  label Time='Survival Time'
        VStatus='0=Alive 1=Dead';
  datalines;
1.25 1 2.2175 9.4 1 67 3.6628 1 1.9542 12 10
1.25 1 1.9395 12.0 1 38 3.9868 1 1.9542 20 18
2.00 1 1.5185 9.8 1 81 3.8751 1 2.0000 2 15
2.00 1 1.7482 11.3 0 75 3.8062 1 1.2553 0 12
2.00 1 1.3010 5.1 0 57 3.7243 1 2.0000 3 9
3.00 1 1.5441 6.7 1 46 4.4757 0 1.9345 12 10
5.00 1 2.2355 10.1 1 50 4.9542 1 1.6628 4 9

... more lines ...

53.00 0 1.1139 12.0 1 66 3.6128 1 2.0000 1 11
57.00 0 1.2553 12.5 1 66 3.9685 0 1.9542 0 11
77.00 0 1.0792 14.0 1 60 3.6812 0 0.9542 0 12
;
```

The variable **Time** represents the survival time in months from diagnosis. The variable **VStatus** indicates whether the patient survived the study. If the value of **VStatus** is 0, then the patient survived, and thus the corresponding value of **Time** is censored. The variables that are thought to be related to survival are **LogBUN** (log of blood urea nitrogen at diagnosis), **HGB** (hemoglobin at diagnosis, in g/dl), **Platelet** (platelets at diagnosis: 0=abnormal, 1=normal), **Age** (age at diagnosis, in years), **LogWBC** (log of white blood cell count at diagnosis), **Frac** (fractures at diagnosis: 0=none, 1=present), **LogPBM** (log percentage of plasma cells in bone marrow), **Protein** (proteinuria at diagnosis, in mg/dl), and **SCalc** (serum calcium at diagnosis, in mg/dl). The goal of the study is to identify important prognostic factors from these nine explanatory variables.

Suppose that T_i represents the survival time of patient i and μ_i represents the RMST at the specified time τ for the patient. A log-linear model for the RMST assumes that

$$\log(\mu_i) = \mathbf{x}_i' \boldsymbol{\beta}$$

The following statements fit the log-linear model for the RMST at $\tau = 50$ with the covariates **LogBUN** and **HGB**:

```
proc rmstreg data=Myeloma tau=50;
  model Time*VStatus(0)=LogBUN HGB / method=ipcw link=log;
run;
```

Figure 21 displays the parameter estimates for the fitted log-linear regression model.

Figure 21 Parameter Estimates
The RMSTREG Procedure

Analysis of Parameter Estimates							
Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
				Intercept	1		
LogBUN	1	-1.0925	0.3053	-1.6908	-0.4942	12.81	0.0003
HGB	1	0.0438	0.0416	-0.0378	0.1254	1.11	0.2923

For comparison, a Cox proportional hazards (PH) model is fitted to the data, using the same set of covariates. The PH model assumes that

$$\lambda(t|\mathbf{x}_i) = \lambda_0(t)\exp(\mathbf{x}_i'\boldsymbol{\beta}_{\text{PH}})$$

where $\lambda_0(t)$ is the baseline hazard function and $\boldsymbol{\beta}_{\text{PH}}$ is a vector of unknown regression parameters.

The following PROC PHREG statements fit a Cox PH model with the covariates **LogBUN** and **HGB**:

```
proc phreg data=Myeloma;
  model Time*VStatus(0)=LogBUN HGB;
run;
```

Figure 22 displays the parameter estimates for the fitted Cox regression model.

Figure 22 Parameter Estimates from Cox Regression
The PHREG Procedure

Analysis of Maximum Likelihood Estimates						
Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
LogBUN	1	1.67440	0.61209	7.4833	0.0062	5.336
HGB	1	-0.11899	0.05751	4.2811	0.0385	0.888

If you compare the results from PROC RMSTREG with the results from PROC PHREG, you see that the estimated effects have opposite signs. This is because Cox regression models the hazard function, and RMST regression models the restricted mean at a certain time via a log link. Thus, you interpret the estimated effects from PROC PHREG as ratios of the hazard functions; you interpret the estimated effects from PROC RMSTREG as log ratios of the RMST. An increase in the hazard rate would correspond roughly to a decrease in the restricted mean, although the exact relationship is difficult to quantify (Karrison 1987).

Also, note that the effect of HGB is significant at the 5% level in the PROC PHREG results but not significant in the PROC RMSTREG results. This might be due to the particular time τ that is specified for the RMST analysis, or it might be due to the different characteristics of PROC PHREG's likelihood-based approach versus PROC RMSTREG's approach, which uses estimating equations.

You can also compare the RMST regression model to the accelerated failure time (AFT) model that is fit by the LIFEREG procedure. The AFT model assumes that

$$\log(T_i) = \mathbf{x}_i\boldsymbol{\beta}_{\text{AFT}} + \sigma\epsilon_i$$

where $\boldsymbol{\beta}_{\text{AFT}}$ is a vector of unknown regression parameters, ϵ_i is the error term that is sampled from a known distribution, and σ is an unknown scale parameter.

The following PROC LIFEREG statements fit the AFT model with a Weibull distribution, using the same set of covariates as in the previous examples:

```
proc lifereg data=Myeloma;
  model Time*VStatus(0)=LogBUN HGB;
run;
```

Figure 23 displays the parameter estimates for the fitted AFT regression model.

Figure 23 Parameter Estimates from AFT Model

The LIFEREG Procedure

Analysis of Maximum Likelihood Parameter Estimates							
Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
				Intercept	1		
LogBUN	1	-1.5304	0.5031	-2.5165	-0.5444	9.25	0.0024
HGB	1	0.0975	0.0492	0.0011	0.1939	3.93	0.0474
Scale	1	0.8770	0.0945	0.7100	1.0833		
Weibull Shape	1	1.1403	0.1229	0.9231	1.4085		

The estimated effects from the AFT model exhibit the same pattern as the estimated effects from the log-linear model of PROC RMSTREG. However, the interpretations of the two models are quite different. Instead of modeling the truncated mean of the event time, as PROC RMSTREG does, the AFT model assumes that the effect of the independent variables on the event time is additive on the log scale. As with the Cox regression model that is fit by PROC PHREG, it is difficult to translate the estimated effects from PROC LIFEREG to the scale of the RMST model.

SUMMARY

The restricted mean survival time (RMST) is an alternative measure to the survival outcome. Compared to the survival mean and median, it can be more robustly estimated and is still valid when the proportional hazards assumption is violated. A time horizon needs to be specified up front, before you perform an RMST analysis, and you should decide how to perform the analysis on the basis of the analytical objective and meaningful interpretations. In certain applications, the largest observed time is not a bad choice, because it would use most of the survival information.

New features available in SAS/STAT 15.1 enable you to perform different types of statistical analysis of the RMST. You can use the new RMSTREG procedure to perform regression analysis and the new RMST option in PROC LIFETEST to perform nonparametric analyses. These RMST-based methods strongly parallel their classical counterparts. Nonparametric analyses are concerned with estimating the RMST and comparing the RMST between groups. You can consider the RMST option in PROC LIFETEST to be a bonus option because it performs RMST analysis in addition to the existing functionality in PROC LIFETEST. Regression analysis aims to study the relationship between the RMST and independent variables. PROC RMSTREG fits linear and log-linear models of the RMST, and these models provide a useful alternative to the classical proportional hazards and accelerated failure time models.

REFERENCES

- Andersen, P. K., Hansen, M. G., and Klein, J. P. (2004). "Regression Analysis of Restricted Mean Survival Time Based on Pseudo-observations." *Lifetime Data Analysis* 10:335–350.
- Andersen, P. K., Klein, J. P., and Rosthøj, S. (2003). "Generalised Linear Models for Correlated Pseudo-observations, with Applications to Multi-state Models." *Biometrika* 90:15–27.
- Hosmer, D. W., Jr., and Lemeshow, S. (1999). *Applied Survival Analysis: Regression Modeling of Time-to-Event Data*. New York: John Wiley & Sons.
- Kalbfleisch, J. D., and Prentice, R. L. (1980). *The Statistical Analysis of Failure Time Data*. New York: John Wiley & Sons.

- Karrison, T. (1987). "Restricted Mean Life with Adjustment for Covariates." *Journal of the American Statistical Association* 82:1169–1176.
- Krall, J. M., Uthoff, V. A., and Harley, J. B. (1975). "A Step-Up Procedure for Selecting Variables Associated with Survival." *Biometrics* 31:49–57.
- Nelder, J. A., and Wedderburn, R. W. M. (1972). "Generalized Linear Models." *Journal of the Royal Statistical Society, Series A* 135:370–384.
- Royston, P., and Parmar, M. K. B. (2013). "Restricted Mean Survival Time: An Alternative to the Hazard Ratio for the Design and Analysis of Randomized Trials with a Time-to-Event Outcome." *BMC Medical Research Methodology* 13:152–166.
- Tian, L., Zhao, L., and Wei, L. J. (2014). "Predicting the Restricted Mean Event Time with the Subject's Baseline Covariates in Survival Analysis." *Biostatistics* 15:222–233.
- Trinquart, L., Jacot, J., Conner, S. C., and Porcher, R. (2016). "Comparison of Treatment Effects Measured by the Hazard Ratio and by the Ratio of Restricted Mean Survival Times in Oncology Randomized Controlled Trials." *Journal of Clinical Oncology* 34:1813–1819.
- Uno, H., Claggett, B., Tian, L., Inoue, E., Gallo, P., Miyata, T., Schrag, D., et al. (2014). "Moving Beyond the Hazard Ratio in Quantifying the Between-Group Difference in Survival Analysis." *Journal of Clinical Oncology* 32:2380–2385.
- Zucker, D. M. (1998). "Restricted Mean Life with Covariates: Modification and Extension of a Useful Survival Analysis Method." *Journal of the American Statistical Association* 93:702–709.

ACKNOWLEDGMENTS

The authors thank Ed Huddleston for his valuable editorial assistance in preparing this paper. The authors also thank Bob Rodriguez and Maura Stokes, formerly of the Advanced Analytics Division at SAS, for their leadership and support in the development of the software discussed in the paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors:

Changbin Guo	Yu Liang
SAS Institute Inc.	SAS Institute Inc.
SAS Campus Drive	SAS Campus Drive
Cary, NC 27513	Cary, NC 27513
changbin.guo@sas.com	yu.liang@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.